// 24.07.2020, Arduino IDE v1.8.9, v0.2.1

// CW Beacon using Si5351: CLK0, CLK1, CLK2, LZ2WSG, KN34PC

//---------------------------------------------------------------------------------------------------------




#define PIN_SP 10//pin 10 arduino uscita monitor manipolazione cw

#define PIN_TX 13//pin 13 arduino uscita +vcc led monitor manipolazion


uint16_t duration = 120;//velocita' manipolazione cw (valore alto=velocita' bassa...valore basso=velocita' alta)

uint16_t hz = 750;

String cw_message = "TESTO DEL MESSAGGIO"; // inserire il msg. beacon

uint32_t tx_1 = 20000;    // CLK1   // inserire frequenza voluta (es.50.400 mhz= 50400000)

uint32_t tx_2 = 20000;   // CLK2  //  inserire frequenza voluta (es. 144.400= 144400000)


//---------------------------------------------------------------------------------------------------------

void setup() {

  pinMode(PIN_TX, OUTPUT);


  si5351.init(SI5351_CRYSTAL_LOAD_8PF, 25000000, 0);// per divari di frequenza out correggere valore quarzo

  si5351.drive_strength(SI5351_CLK1, SI5351_DRIVE_8MA);//pwr out si5351....2MA=2mW......8MA=8mW

  si5351.set_freq(tx_1 * SI5351_FREQ_MULT, SI5351_CLK1);

  si5351.output_enable(SI5351_CLK1, 0);

  si5351.drive_strength(SI5351_CLK2, SI5351_DRIVE_8MA);//pwr out si5351

  si5351.set_freq(tx_2 * SI5351_FREQ_MULT, SI5351_CLK2);

  si5351.output_enable(SI5351_CLK2, 0);


  

}

//------------------------------TEMPI ON/OFF BEACON ESPRESSI IN MILLISECONDI....1000 MILLIS=1 SECONDO-----------

void loop() {

  cw_string_proc(cw_message);

  delay(1500);// tempo ritardo tra msg. e portante fissa (>1000)


  cw(true);

  delay(18000);//tempo della portante fissa (>12000)


  cw(false);

  delay(10000);//tempo di bianco prima del loop (10 sec.) per monitorare altri bcn isofrequenza(>10000)

}

//---------------------------------------------------------------------------------------------------------

void cw_string_proc(String str) {                      // processing string to characters

  for (uint8_t j = 0; j < str.length(); j++)

    cw_char_proc(str[j]);

}

//---------------------------------------------------------------------------------------------------------

void cw_char_proc(char m) {                            // processing characters to Morse code symbols

  String s;


  if (m == ' ') {                                      // pausa tra le parole

    word_space();

    return;

  }


  if (m > 96)                                          // ACSII, case change a-z to A-Z

    if (m < 123)

      m -= 32;


  switch (m) {                                         // Mcodice morse

    case 'A': s = ".-#";     break;

    case 'B': s = "-...#";   break;

    case 'C': s = "-.-.#";   break;

    case 'D': s = "-..#";    break;

    case 'E': s = ".#";      break;

    case 'F': s = "..-.#";   break;

    case 'G': s = "--.#";    break;

    case 'H': s = "....#";   break;

    case 'I': s = "..#";     break;

    case 'J': s = ".---#";   break;

    case 'K': s = "-.-#";    break;

    case 'L': s = ".-..#";   break;

    case 'M': s = "--#";     break;

    case 'N': s = "-.#";     break;

    case 'O': s = "---#";    break;

    case 'P': s = ".--.#";   break;

    case 'Q': s = "--.-#";   break;

    case 'R': s = ".-.#";    break;

    case 'S': s = "...#";    break;

    case 'T': s = "-#";      break;

    case 'U': s = "..-#";    break;

    case 'V': s = "...-#";   break;

    case 'W': s = ".--#";    break;

    case 'X': s = "-..-#";   break;

    case 'Y': s = "-.--#";   break;

    case 'Z': s = "--..#";   break;


    case '1': s = ".----#";  break;

    case '2': s = "..---#";  break;

    case '3': s = "...--#";  break;

    case '4': s = "....-#";  break;

    case '5': s = ".....#";  break;

    case '6': s = "-....#";  break;

    case '7': s = "--...#";  break;

    case '8': s = "---..#";  break;

    case '9': s = "----.#";  break;

    case '0': s = "-----#";  break;


    case '?': s = "..--..#"; break;

    case ',': s = "--..--#"; break;

    case '/': s = "-..-.#";  break;

  }


  for (uint8_t i = 0; i < 7; i++) {

    switch (s[i]) {

      case '.': ti();  break;                          // punto

      case '-': ta();  break;                          // linea

      case '#': char_space(); return;                  // fine codice morse

    }

  }

}

//---------------------------------------------------------------------------------------------------------

void ti() {

  cw(true);                                            // trasmissione punto

  delay(duration);


  cw(false);                                           // stop trasmissione punto

  delay(duration);

}

//---------------------------------------------------------------------------------------------------------

void ta() {

  cw(true);                                            // trasmissione linea

  delay(3 * duration);


  cw(false);                                           // stop trasmissione linea

  delay(duration);

}

//---------------------------------------------------------------------------------------------------------

void char_space() {                                    // 3x, pausa tra le lettere

  delay(2 * duration);                                 // 1 from element-end + 2 new

}

//---------------------------------------------------------------------------------------------------------

void word_space() {                                    // 7x, pausa tra le parole

  delay(6 * duration);                                 // 1 from element-end + 6 new

}

//---------------------------------------------------------------------------------------------------------

void cw(bool state) {                                  // TX-CW, TX-LED, 750 Hz sound

  if (state) {

    si5351.output_enable(SI5351_CLK1, 1);

    si5351.output_enable(SI5351_CLK2, 1);

    digitalWrite(PIN_TX, HIGH);

    tone(PIN_SP, hz);

  }

  else {

    si5351.output_enable(SI5351_CLK1, 0);

    si5351.output_enable(SI5351_CLK2, 0);

    digitalWrite(PIN_TX, LOW);

    noTone(PIN_SP);

  }

}

